Uzawa-Lucas model (complete)

(Quantitative Dynamic Macroeconomics, Lecture Notes, Thomas Steger, University of Leipzig)

1. The model

Introduction

In this section we consider a human-capital-based endogenous growth model. The standard approach is due to Uzawa (1965) and Lucas (1988) and is hence sometimes called Uzawa-Lucas model. We focus on a human-capital-based growth model as formulated by Benhabib and Perli (1994).

Final output is produced by employing physical capital and human capital. The accumulation of human capital uses a separate technology. It is assumed that the production of human capital is human-capital intensive. In fact, human capital is the sole input factor used in the education sector. This formulation can be considered as an approximation of the assumption according to which education is human capital intensive. It is further assumed that the technology available to the education sector exhibits constant returns to scale. This last assumption is, of course, critical for the generation of sustained growth.

The model

The dynamic problem of the representative agent may be expressed as follows

$$\begin{aligned} &\max_{\{C,u\}} \int_0^\infty \frac{C^{1-\sigma}-1}{1-\sigma} e^{-\rho t} dt \\ &s.t.\ \dot{K} = A K^\alpha (u H)^{1-\alpha} H_a^\gamma - C \qquad \text{with } A, \gamma > 0 \text{ and } 0 < \alpha < 1 \\ &\dot{H} = \beta H (1-u) \qquad \text{with } \beta > 0 \\ &K(0) = K_0 \\ &H(0) = H_0 \end{aligned} \tag{1}$$

COMMENTS

[1] This is the problem of the representative agent (producer-consumer household). Therefore, the average stock of human capital H_a is taken as exogenous. The social planner would consider H_a as being endogenous.

[2] There are two control variables, namely the rate of consumption C and the time allocation variable u. Moreover, there are two state variables, the stock of physical capital K and the stock of human capital H.

[3] Notice that there is a positive externality associated with the average stock of human capital H_a in final output production. This positive externality is, however, not necessary to generate sustained growth.

First-order conditions

As usual, we set up the current-value Hamiltonian

```
Off[General::spell]
Off[General::spell1]
```

```
Clear[HH, obj, res1, res2];
obj = (C^(1-σ) - 1)/(1 - σ); res1 = A K^α (u H)^(1-α) Ha^γ - C ; res2 = β H (1 - u);
HH = obj + λK res1 + λH res2;
```

and then derive the necessary first-order condition based on the Hamiltonian set up above

```
foc1 = ∂_C HH == 0;
foc2 = ∂_u HH == 0;
foc3 = λK ' == -∂_K HH + ρ λK;
foc4 = λH ' == -∂_H HH + ρ λH;
foc5 = K ' == ∂_λK HH;
foc6 = H ' == ∂_λH HH;
(listFoc = {foc1, foc2, foc3, foc4, foc5, foc6}) // ColumnForm
```

```
C^-σ - λK == 0
-H β λH + A H Ha^γ K^α (H u)^-α (1 - α) λK == 0
λK′ == -A Ha^γ K^(-1+α) (H u)^(1-α) α λK + λK ρ
λH′ == -(1 - u) β λH - A Ha^γ K^α u (H u)^-α (1 - α) λK + λH ρ
K′ == -C + A Ha^γ K^α (H u)^(1-α)
H′ == H (1 - u) β
```

In addition, the following transversality conditions must hold

$$\lim_{t\to\infty} e^{-\rho t}\,\lambda_K(t)\,K(t) = 0 \qquad \Rightarrow \qquad -\rho + \lim_{t\to\infty}\hat{\lambda}_K(t) + \lim_{t\to\infty}\hat{K}(t) < 0 \tag{2}$$

$$\lim_{t\to\infty} e^{-\rho t}\,\lambda_H(t)\,H(t) = 0 \qquad \Rightarrow \qquad -\rho + \lim_{t\to\infty}\hat{\lambda}_H(t) + \lim_{t\to\infty}\hat{H}(t) < 0 \tag{3}$$

COMMENTS

[1] For remarks on the TVC and the sufficiency conditions see Benhabib and Perli (1994, pp. 117/118).

[2] As usual, the differential equation in C (the KRR) results when FOC1 is differentiated w.r.t. time and $\dot{\lambda}$ is substituted according to FOC3.

Eliminating shadow prices

By noting that, in equilibrium, $H = H_a$ FOC2 can be written as

$$A\,H^{1-\alpha+\gamma}\,K^{\alpha}\;u^{-\alpha}\,(1-\alpha)\,\lambda_K - H\,\beta\,\lambda_H = 0 \tag{4}$$

Written in growth rates (this equation will be used to derive a DE in u)

$$(1-\alpha+\gamma)\,\hat{H} + \alpha\,\hat{K} - \alpha\,\hat{u} + \hat{\lambda}_K - \hat{H} - \hat{\lambda}_H = 0 \tag{5}$$

The growth rates $\hat{H}$, $\hat{K}$, $\hat{\lambda}_K$, and $\hat{\lambda}_H$ can be substituted according to

$$\hat{H} = \beta(1-u) \tag{6}$$

$$\hat{K} = A\,K^{\alpha-1}(u\,H)^{1-\alpha}\,H_a^{\gamma} - C/K \tag{7}$$

$$\hat{\lambda}_K = \rho - A\,K^{-1+\alpha}\,H^{1-\alpha+\gamma}\,u^{1-\alpha}\,\alpha \tag{8}$$

$$\hat{\lambda}_H = \rho - (1-u)\,\beta\, - A\,H^{\gamma-\alpha}\,K^{\alpha}\;u^{1-\alpha}\,(1-\alpha)\,\frac{\lambda_K}{\lambda_H} = \rho - \beta \tag{9}$$

where the last equation uses $\frac{\lambda_K}{\lambda_H} = \frac{H\,\beta}{A\,K^{\alpha}\,H^{-\alpha+\gamma+1}\,u^{-\alpha}\,(1-\alpha)}$ (from FOC2)

Hence, equ. (5) may be expressed as

$$(1-\alpha+\gamma)\,\beta\,(1-u) + \alpha\left(A\,K^{\alpha-1}\,u^{1-\alpha}\,H^{1-\alpha+\gamma} - C/K\right) - \\ \alpha\,\hat{u} + \rho - A\,K^{-1+\alpha}\,H^{1-\alpha+\gamma}\,u^{1-\alpha}\,\alpha\, - \beta(1-u) - \rho + \beta = 0 \tag{10}$$

$$(1-\alpha+\gamma)\,\beta\,(1-u) - \alpha\,C/K - \alpha\,\hat{u}\, - \beta(1-u) + \beta = 0 \tag{11}$$

$$(1-\alpha+\gamma)\,\beta\,(1-u) - \alpha\,C/K\, + \beta\,u = \alpha\,\hat{u} \tag{12}$$

$$\alpha\,\hat{u} = (1-\alpha+\gamma)\,\beta\,(1-u) - \alpha\,C/K\, + \beta\,u \tag{13}$$

$$\dot{u} = u\left(\frac{(1-\alpha+\gamma)\,\beta}{\alpha}\,(1-u) - C/K\, + \frac{\beta}{\alpha}\,u\right) \tag{14}$$

Complete dynamic system

The complete dynamic system in K, H, C, and u reads as follows (boundary conditions suppressed)

$$\dot{K} = A\;K^{\alpha}\;H^{1-\alpha+\gamma}\,u^{1-\alpha} - C \tag{15}$$

$$\dot{H} = \beta\,(1-u)\,H \tag{16}$$

$$\dot{C} = \frac{C}{\sigma}\left(\alpha\,A\;K^{\alpha-1}\;H^{1-\alpha+\gamma}\,u^{1-\alpha} - \rho\right) \tag{17}$$

$$\dot{u} = u\left(\frac{(\gamma-\alpha)\,\beta}{\alpha}\,(1-u) + \frac{\beta}{\alpha} - \frac{C}{K}\right) \tag{18}$$

Steady state growth rates are given by

$$g_K = g_C = g = \frac{1-\alpha+\gamma}{(1-\alpha+\gamma)\,\sigma - \gamma}\,(\beta-\rho) \tag{19}$$

$$g_H = \frac{1-\alpha}{1-\alpha+\gamma}\,g \tag{20}$$

$$g_u = 0 \tag{21}$$

2. Model analysis

The dynamic system in scale-adjusted variables

Define $k := K\, e^{-g\, t}$, $h := H\, e^{-g_H\, t}$, $c := C\, e^{-g\, t}$. The dynamic system in scale-adjusted variables then reads

$$\dot{k} = A\, k^{\alpha}\, h^{1-\alpha+\gamma}\, u^{1-\alpha} - c - g_K\, k \tag{22}$$

$$\dot{h} = \beta\, (1-u)\, h - g_H\, h \tag{23}$$

$$\dot{c} = \frac{c}{\sigma}\left(\alpha\, A\, k^{\alpha-1}\, h^{1-\alpha+\gamma}\, u^{1-\alpha} - \rho\right) - g_C\, c \tag{24}$$

$$\dot{u} = u\left(\frac{(\gamma-\alpha)\,\beta}{\alpha}\,(1-u) + \frac{\beta}{\alpha} - \frac{c}{k}\right) \tag{25}$$

$$g_K = g_C = g = \frac{1-\alpha+\gamma}{(1-\alpha+\gamma)\,\sigma-\gamma}\,(\beta-\rho) \tag{26}$$

$$g_H = \frac{1-\alpha}{1-\alpha+\gamma}\, g \tag{27}$$

This dynamic system in scale-adjusted variables does not define a unique stationary solution. To determine an (initial) steady state, we are forced to set one of the endogenous variables at a specific value, say $k(0) = k_0$. This implies that the above system defines a continuum of steady states (a center manifold), i.e. for each values of k_0 there is a different set of steady state $\{\tilde{h}, \tilde{c}, \tilde{u}\}$.

Trying to find a stationary solution (or: the center manifold)

A (nontrivial, i.e. interior) stationary solution of the scale-adjusted dynamic system is a set of values for k, h, c, and u that solves

$$\dot{k}/k = A\, k^{\alpha-1}\, h^{1-\alpha+\gamma}\, u^{1-\alpha} - \delta - c/k - g_K = 0 \tag{28}$$

$$\dot{h}/h = \beta\,(1-u) - g_H = 0 \quad \rightarrow \quad \tilde{u} \tag{29}$$

$$\dot{c}/c = \frac{1}{\sigma}\left(\alpha\, A\, k^{\alpha-1}\, h^{1-\alpha+\gamma}\, u^{1-\alpha} - \delta - \rho\right) - g_C = 0 \tag{30}$$

$$\dot{u}/u = \left(\frac{(\gamma-\alpha)\,\beta}{\alpha}\,(1-u) + \frac{\beta}{\alpha} - \frac{c}{k}\right) = 0 \quad \rightarrow \quad \left(\frac{\tilde{c}}{k}\right) \tag{31}$$

Equ. (33) determines u

$$\tilde{u} = 1 - \frac{g_H}{\beta} \tag{32}$$

Equ. (35) then determines $\left(\frac{\tilde{c}}{k}\right)$

$$\left(\frac{(\gamma-\alpha)\,\beta}{\alpha}\,(1-\tilde{u})+\frac{\beta}{\alpha}-\left(\frac{\tilde{c}}{k}\right)\right)=0 \tag{33}$$

The problem then boils down to solving the following two equations w.r.t. *h* and *k*

$$A\,k^{\alpha-1}\,h^{1-\alpha+\gamma}\,\tilde{u}^{1-\alpha}-\delta-\left(\frac{\tilde{c}}{k}\right)-g_K=0 \tag{34}$$

$$\frac{1}{\sigma}\left(\alpha\,A\,k^{\alpha-1}\,h^{1-\alpha+\gamma}\,\tilde{u}^{1-\alpha}-\delta-\rho\right)-g_C=0 \tag{35}$$

These two equations describe curves in (*k*, *h*)-plane. The intersection(s) of these two curves gives us the remaining solution(s) for *k* and *h*. The phenomenon under study is best understood by plotting these two curves in (*k*, *h*)-plane.

```
<< Graphics`ImplicitPlot`;
```

General::obspkg :
Graphics`ImplicitPlot` is now obsolete. The legacy version being loaded may conflict with current *Mathematica* functionality. See the Compatibility Guide for updating information. ≫

```
paramFinal = {σ → 1.5, α → 0.3, γ → 0.3, β → 0.1, ρ → 0.05, A → 1};
```

```
equ1 = (A k^(α-1) h^(1-α+γ) u^(1-α) - ck - gK == 0) /. u -> (β - gH)/β /. ck -> (β + (γ - α) gH)/α /.
    gH -> (1 - α)/(1 - α + γ) gK /. gK -> (1 - α + γ)/((1 - α + γ) σ - γ) (β - ρ);
equ2 =
  1/σ (α A k^(α-1) h^(1-α+γ) u^(1-α) - ρ) - gC == 0 /. u -> (β - gH)/β /. ck -> (β + (γ - α) gH)/α /.
    gH -> (1 - α)/(1 - α + γ) gK /. gC → gK /. gK -> (1 - α + γ)/((1 - α + γ) σ - γ) (β - ρ);
curve1 = ImplicitPlot[equ1 /. paramFinal, {k, 0.0001, 1},
   {h, 0.0001, 1}, AspectRatio → 1, AxesLabel → {"k", "h"},
   PlotRange → {{0, 1}, {0, 0.7}}, PlotLabel → "equ.(35)"];
curve2 = ImplicitPlot[equ2 /. paramFinal, {k, 0.0001, 1},
   {h, 0.0001, 0.8}, AspectRatio → 1, AxesLabel → {"k", "h"},
   PlotRange → {{0, 1}, {0, 0.7}}, PlotLabel → "equ.(36)"];
GraphicsArray[
  {{curve1, curve2, Show[curve1, curve2, PlotLabel → "equ.(35)&(36)"]}}]
```



The two curves coincide. This means that there is a continuum of stationary solutions. The above curve illustrates the center manifold. More precisely, the above displayed curves illustrates (the projection of) the center manifold in (k, h)-plane.

■ Some manual checks

3. Simulating the transition process

The dynamic system in scale-adjusted variables (once more)

$$\dot{k} = A\, k^{\alpha}\, h^{1-\alpha+\gamma}\, u^{1-\alpha} - c - g_K\, k \tag{42}$$

$$\dot{h} = \beta\,(1-u)\,h - g_h\, h \tag{43}$$

$$\dot{c} = \frac{c}{\sigma}\left(\alpha\, A\, k^{\alpha-1}\, h^{1-\alpha+\gamma}\, u^{1-\alpha} - \rho\right) - g_C\, c \tag{44}$$

$$\dot{u} = u\left(\frac{(\gamma-\alpha)\,\beta}{\alpha}\,(1-u) + \frac{\beta}{\alpha} - \frac{c}{k}\right) \tag{45}$$

$$g_K = g_C = g = \frac{1-\alpha+\gamma}{(1-\alpha+\gamma)\,\sigma-\gamma}\,(\beta-\rho) \tag{46}$$

$$g_H = \frac{1-\alpha}{1-\alpha+\gamma}\, g \tag{47}$$

Here we define the dynamic system (scale-adjusted variables) in *Mathematica* syntax

```
{g, gH} = {(1 - α + γ)/((1 - α + γ) σ - γ) (β - ρ), (1 - α)/(1 - α + γ) (1 - α + γ)/((1 - α + γ) σ - γ) (β - ρ)};
```

```
dynEqu = {A k^α h^(1-α+γ) u^(1-α) - c - g k, β (1 - u) h - gH h,
   c/σ (α A k^(α-1) h^(1-α+γ) u^(1-α) - ρ) - g c, u ((γ - α) β/α (1 - u) + β/α - c/k)};
```

```
paramFinal = {σ → 1.5, α → 0.3, γ → 0.3, β → 0.1, ρ → 0.05, A → 1};
```

```
{g, gH, 1 - (1 - α)/((1 - α + γ) σ - γ) (1 - ρ/β)} /. paramFinal
```

```
{0.0416667, 0.0291667, 0.708333}
```

An initial steady state

To guarantee that the initial steady state is not too far away from the final steady state, one can set $k = 1$, say, then calculate $\tilde{h}(k = 1)$ and $\tilde{c}(k = 1)$ (notice that u^* is independent of k). Then reduce k by a small percentage, such that the initial steady state is not the new steady state.

```
solh = Solve[k^(α-1) h^(1-α+γ) == (σ g + ρ)/(α A) u^(α-1) /. u -> 1 - gH/β, h];
```

Solve::ifun :
Inverse functions are being used by Solve, so some solutions may not be found;
use Reduce for complete solution information. ≫

```
hInitial = solh[[1, 1, 2]] /. k → 1 // Simplify;
```

```
{uInitial, cInitial, kInitial} = {1 - gH/β, ((γ - α) (1-α)/(1-α+γ) g + β)/α kInitial, 0.5};
```

Generic notation

Here we switch to a generic notation by defining and applying a substitution list. X-type variables denote dynamic state variables; W-type variables denote dynamic jump variables; Y-type variables denote jump variables, which are determined by static equations.

```
subsVar = {k -> X[1, j], h -> X[2, j], c -> W[1, j], u -> W[2, j]};
```

```
dynEquScale = dynEqu /. subsVar;
```

Discretization: algebraic system

```
{nX, nW, nY} = {2, 2, 0}(*number of different type of variables*);

n = 75 (*number of mesh points*);

subsList1 = Join[Table[X[i, j] → ((X[i, j - 1] + X[i, j])/2), {i, 1, nX}],
   Table[W[i, j] → ((W[i, j - 1] + W[i, j])/2), {i, 1, nW}]];

equMain =
  Table[Join[Table[X[i, j] - X[i, j - 1], {i, 1, nX}],
     Table[W[i, j] - W[i, j - 1], {i, 1, nW}]] - (dynEquScale /. subsList1),
   {j, 1, n}];(*discretization: (nX+nW)*n equations*)

equBorder = Join[{X[1, 0] - kInitial, X[2, 0] - hInitial /. paramFinal},
   Drop[dynEquScale /. j → n, {1, 2}]];
 (*border equations: nX+nW equations*)

equations = Join[Flatten[equMain], equBorder];
```

Rootfinding

```
start1 = {{X[1, j], kInitial}, {X[2, j], hInitial /. paramFinal},
   {W[1, j], cInitial /. paramFinal}, {W[2, j], uInitial /. paramFinal}};
startValues = Flatten[Table[start1, {j, 0, n}], 1];
```

```
Timing[sol2 = FindRoot[equations /. paramFinal, startValues];]
Max[(equations /. paramFinal) /. sol2]
```

```
{0.203, Null}
```

2.77556×10^{-16}

```
p1 = ListPlot[Table[sol2[[i, 2]], {i, 1, (nX + nW + nY) n, nX + nW + nY}],
  AxesLabel → {t, k}]
p2 = ListPlot[Table[sol2[[i, 2]], {i, 2, (nX + nW + nY) n, nX + nW + nY}],
  AxesLabel → {t, h}]
p3 = ListPlot[Table[sol2[[i, 2]], {i, 3, (nX + nW + nY) n, nX + nW + nY}],
  AxesLabel → {t, c}]
p3 = ListPlot[Table[sol2[[i, 2]], {i, 4, (nX + nW + nY) n, nX + nW + nY}],
  AxesLabel → {t, u}]
```










```
kList = Table[sol2[[i, 2]], {i, 1, (nX + nW + nY) n, nX + nW + nY}];
hList = Table[sol2[[i, 2]], {i, 2, (nX + nW + nY) n, nX + nW + nY}];
```

```
curve = Thread[{kList, hList}];
```

```
trajectory = ListPlot[curve, PlotRange → {{0, 2}, {0, 2}},
   AxesLabel → {"k", "h"}];
Show[curve1, trajectory,
 PlotLabel → "convergence to the center manifold"]
```



Notice that the final steady state cannot be determined (not even numerically) without calculating the equilibrium trajectory. It results from the transition process and hence depends on initial conditions ($k(0)$ and $h(0)$). From each point in (k, h)-plane there is a unique trajectory converging to a specific point on the center manifold. The resulting point on the center manifold determines the level of the BGP. Consequently, the level of the BGP (in original variables) depends on initial conditions (this statement should not come at a surprise).

4. Stability properties of the center manifold (eigenvalues)

The dynamic system in scale-adjusted variables (for convenience)

$$\dot{k} = A\, k^{\alpha}\, h^{1-\alpha+\gamma}\, u^{1-\alpha} - \delta\, k - c - g_K\, k \tag{48}$$

$$\dot{h} = \beta\,(1-u)\,h - g_H\,h \tag{49}$$

$$\dot{c} = \frac{c}{\sigma}\left(\alpha\,A\,k^{\alpha-1}\,h^{1-\alpha+\gamma}\,u^{1-\alpha} - \delta - \rho\right) - g_C\,c \tag{50}$$

$$\dot{u} = u\left(\frac{(\gamma-\alpha)\,\beta}{\alpha}\,(1-u) + \frac{\beta}{\alpha} - \frac{c}{k}\right) \tag{51}$$

$$g_K = g_C = g = \frac{1-\alpha+\gamma}{(1-\alpha+\gamma)\,\sigma-\gamma}\,(\beta-\rho) \tag{52}$$

$$g_H = \frac{1-\alpha}{1-\alpha+\gamma}\,g_K \tag{53}$$

```
solh = Solve[k^(α-1) h^(1-α+γ) == (σ g + ρ)/(α A) u^(α-1) /. u -> 1 - gH/β, h] // Simplify
```

Solve::ifun :

Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information. ≫

$$\left\{\left\{h \to \left(\frac{k^{1-\alpha}\,\beta\,\left(1 + \frac{(-1+\alpha)\,(\beta-\rho)}{\beta\,(\gamma\,(-1+\sigma)+\sigma-\alpha\,\sigma)}\right)^{\alpha}\,((-1+\alpha)\,\beta\,\sigma + \gamma\,(\rho-\beta\,\sigma))}{A\,\alpha\,((-1+\alpha)\,\rho + \beta\,(-1+\alpha-\gamma)\,(-1+\sigma))}\right)^{\frac{1}{1-\alpha+\gamma}}\right\}\right\}$$

```
{g, gH} = {(1 - α + γ)/((1 - α + γ) σ - γ) (β - ρ), (1 - α)/(1 - α + γ) (1 - α + γ)/((1 - α + γ) σ - γ) (β - ρ)};
```

```
{kdot, hdot, cdot, udot} =
  {A k^α h^(1-α+γ) u^(1-α) - c - g k, β (1 - u) h - gH h, c/σ (α A k^(α-1) h^(1-α+γ) u^(1-α) - ρ) - g c,
   u ((γ - α) β/α (1 - u) + β/α - c/k)};
```

```
(jac1 = D[{kdot, hdot, cdot, udot}, {{k, h, c, u}}]) // TraditionalForm
```

$$\begin{pmatrix} A\,h^{-\alpha+\gamma+1}\,k^{\alpha-1}\,u^{1-\alpha}\,\alpha - \frac{(-\alpha+\gamma+1)\,(\beta-\rho)}{(-\alpha+\gamma+1)\,\sigma-\gamma} & A\,h^{\gamma-\alpha}\,k^{\alpha}\,u^{1-\alpha}\,(-\alpha+\gamma+1) & -1 \\ 0 & (1-u)\,\beta - \frac{(1-\alpha)\,(\beta-\rho)}{(-\alpha+\gamma+1)\,\sigma-\gamma} & 0 \\ \frac{A\,c\,h^{-\alpha+\gamma+1}\,k^{\alpha-2}\,u^{1-\alpha}\,(\alpha-1)\,\alpha}{\sigma} & \frac{A\,c\,h^{\gamma-\alpha}\,k^{\alpha-1}\,u^{1-\alpha}\,\alpha\,(-\alpha+\gamma+1)}{\sigma} & \frac{A\,h^{-\alpha+\gamma+1}\,k^{\alpha-1}\,u^{1-\alpha}\,\alpha-\rho}{\sigma} - \frac{(-\alpha+\gamma+1)\,(\beta-\rho)}{(-\alpha+\gamma+1)\,\sigma-\gamma} \\ \frac{c\,u}{k^2} & 0 & -\frac{u}{k} \end{pmatrix}$$

```
(jac2 = jac1 /. {u → 1 - gH/β, c -> ((γ - α) (1-α)/(1-α+γ) g + β)/α k} /.
    h → solh[[1, 1, 2]]) // TraditionalForm
```

$$\begin{pmatrix} A\,k^{\alpha-1}\,\alpha\left(\left(\frac{k^{1-\alpha}\,\beta\left(\frac{(\alpha-1)(\beta-\rho)}{\beta(\gamma(\sigma-1)-\alpha\,\sigma+\sigma)}+1\right)^{\alpha}((\alpha-1)\,\beta\,\sigma+\gamma(\rho-\beta\,\sigma))}{A\,\alpha((\alpha-1)\,\rho+\beta(\alpha-\gamma-1)(\sigma-1))}\right)^{\frac{1}{-\alpha+\gamma+1}}\right)^{-\alpha+\gamma+1}\left(1-\frac{(1-\alpha)(\beta-\rho)}{\beta((-\alpha+\gamma+1)\,\sigma-\gamma)}\right)^{1-\alpha}-\frac{(-\alpha+\gamma+1)(\beta}{(-\alpha+\gamma+1)\,\sigma} \\ 0 \\ \frac{A\,k^{\alpha-1}(\alpha-1)\left(\left(\frac{k^{1-\alpha}\,\beta\left(\frac{(\alpha-1)(\beta-\rho)}{\beta(\gamma(\sigma-1)-\alpha\,\sigma+\sigma)}+1\right)^{\alpha}((\alpha-1)\,\beta\,\sigma+\gamma(\rho-\beta\,\sigma))}{A\,\alpha((\alpha-1)\,\rho+\beta(\alpha-\gamma-1)(\sigma-1))}\right)^{\frac{1}{-\alpha+\gamma+1}}\right)^{-\alpha+\gamma+1}\left(1-\frac{(1-\alpha)(\beta-\rho)}{\beta((-\alpha+\gamma+1)\,\sigma-\gamma)}\right)^{1-\alpha}\left(\beta+\frac{(1-\alpha)(\gamma-\alpha)(\beta-}{(-\alpha+\gamma+1)\,\sigma-}\right.}{\sigma} \\ \frac{\left(1-\frac{(1-\alpha)(\beta-\rho)}{\beta((-\alpha+\gamma+1)\,\sigma-\gamma)}\right)\left(\beta+\frac{(1-\alpha)(\gamma-\alpha)(\beta-\rho)}{(-\alpha+\gamma+1)\,\sigma-\gamma}\right)}{k\,\alpha} \end{pmatrix}$$

```
paramFinal = {σ → 1.5, α → 0.3, γ → 0.3, β → 0.1, ρ → 0.05, A → 1};
```

```
eigen = Eigenvalues[jac2 /. paramFinal];
```

```
Table[eigen[[i]], {i, 1, 4}] /. k → 100
```

```
{-0.299815, 0.288858, 0.0817901, 0}
```

```
Plot[Table[eigen[[i]], {i, 1, 4}], {k, 0, 100},
 AxesLabel → {k, eigenvalues}]
```



There are two positive ("unstable") eigenvalues and one negative ("stable") eigenvalue. One eigenvalue is zero, which results from scale adjustment. Notice that the eigenvalues do not seem to change as *k* changes (i.e. the stability properties along the center manifold remain the same). Since there are two jump variables and two unstable eigenvalues, the center manifold is (locally) saddle-point stable.